

Maple for Differential Equations

Author: Fei Ye

Institute: QCC-CUNY

Date: 2021-10-05

Contents

Preface	1
1 Basics in Maple	2
2 Introduction to Differential Equations	13
3 First Order Differential Equations	17
4 Linear Second Order Differential Equations	21

Preface

This is a book written for Maple labs for differential equations.

The source of this book can be found at <https://github.com/fyemath/maple4ode>.

Comments and suggestions are very welcome.

This work is licensed under a Creative Commons Attribution-NonCommercial-ShareAlike 4.0 International License.




by-nc-sa license icon

Chapter 1 Basics in Maple

1.1 Getting Started

When Maple (say Maple2021) starts, you will see the following Maple Start document.




Maple start page screenshot

- The palettes of Maple found on the left side of the Maple window contains expressions and symbols that you can use to quickly enter them.
- The context panel of Maple found on the right side of the Maple window can be used to perform a wide variety of operations on an expression or its output.


If you already know what you want to do, then you may open a new document by clicking New Document icon in the start document. The following shows what a new (empty document) looks like.

- In the context bar of this new document, the current mode is indicated.
- Initially, the Text mode is in use. You may switch to another mode by clicking one of the three modes: Text, Nonexecutable math, or Math.
- Alternatively, you may use the F5 shortcut key, to toggle between these three modes in sequence: text entry, nonexecutable math entry, and executable math.


If you want to explore some featured sample documents, you may go to Start.mw document and click on different icons to open a new document.



Maple new document page screenshot




Different Modes in Maple

- You may always reopen the start page by clicking the home icon  located in the Toolbar to reopen the start page.

To seek help in Maple is easy, in the Math mode, type in the keywords after the question mark ? and press ENTER, you will see a new window popping out with searched results.

- For example, typing in ?solve and pressing ENTER will open the following window.



Example of using the question mark to get quick help in Maple

- Using the ENTER key, the result will appear in the next line.
- To get the result in the SAME line, you may use ALT+ENTER.

In the coming sections, some basics of Maple will be introduced. Another good place to start learning maple is the Maple Quick Reference Card.

1.2 Operators, Variables, and Delimiters

1.2.1 Basic Operators

Use the command `?operators`, you may find descriptions of arithmetic operators in Maple.

	addition	subtraction	multiplication	division	exponentiation
Maple Operators	+	-	*	/	^
In writing	$x + 2$	$a - b$	$2x$	$\frac{p}{q}$	b^5
In Maple	<code>x+2</code>	<code>a-b</code>	<code>2*x</code>	<code>p/q</code>	<code>b^5</code>

In the case of a number multiplied by a variable, the multiplication symbol may be omitted. In general, you can use `*` or a `[space]` to denote multiplication. However, it's highly recommended to use `*` which is easier to debug.

Among all operators in Maple, we will frequently use the assignment operator `:=`. You will see examples in the next section.

1.2.2 Variables

Variables in Maple can be defined using combinations of letters, digits, and underscores, but not beginning with a digit. For example, we frequently use letters as well as letters followed by a number as variable names. Words connected by underscores are also frequently used as variable names.

Note that there are reserved combinations. Those combinations are not allowed in Maple. For example, if you use `sin` as a variable name and try to assign 1 to it using the command `sin:=1`. You will see the following error message

Error, attempting to assign to 'sin' which is protected.
Try declaring 'local sin'; see ?protect for details.

However, you will find that `sinx` is a valid variable name. If you assign 1 to it using the command `sinx:=1`. Pressing ENTER, you will get

$\text{sinx} := 1.$

Sometime you may want to clear the value assigned to a variable. To do so, one way is to assign to the variable its own name:

```
sinx:='sinx'
```

Another way is to use the unassign command `unassign()`:

```
x:=1;
unassign('x');
```

If you would like to forget all previous commands and results, the `restart` command can be used to clear Maple's memory so that it will act (almost) as if just started.

 **Exercise 1.1** Define a variable, assign a number to it, then clear the value assigned to it.

1.2.3 Delimiters

In Maple, commands or functions obey the function notation: `FunctionName()`. Note that there should be no space between the command/function and the parentheses (). For example, `unassign(f)` is valid but `unassign (f)` is not.

Square brackets `[,]` can be used to enclose a list of ordered objects. They can also be used for subscripts. For example, you may define a order list `v` and use `v[1]` to get the first value. Square brackets sometime are also used for options of commands.

Curly brackets `{,}` are used to enclose sets of objects whose order is unimportant.

Triangle brackets `<, >` or `<|>` can be used to create column or row vectors, or matrices.

Examples of usage of above mentioned delimiters can be found in later sections.

1.2.4 Statement Separator


In Maple you may use the semicolon `(;)` and the colon `(:)` to end a statement.

- The semicolon is the normal statement separator.
- When using the colon, the statement will be executed but the result of the statement will not be displayed.
- Statement separator may be omitted if there is only one statement in a single line.

Example 1.1

Entering the following commands will only display the values of b , c and d but not a

```
a := 1:
b := a+1; c:=b+1;
d := c+1
```

 **Exercise 1.2** Define a variable and assign a value to it without display the result. Add the value 1 to the variable and display the result.

1.3 Functions and Evaluation

1.3.1 How to define a function

A function is an assignment, for a given input x , we assign an output y under a certain rule. Maple takes this idea to define functions.

`function name := independent variable -> function rule`

Here `:=` means “defined/assigned to be” and the arrow operator `->` may be understood as “plug in”.

Another way to define a function is using the function notation as follows

`functionname(independent variable) := expression`

Note that this feature is only available in Maple 10 or a later version.

Yet another uncommon, but useful way to define a function is to use the command `unapply`. This command turns an expression into an operator.

Example 1.2 Define the following function in Maple and find the value $f(0.999)$.

$$f(x) = \frac{x}{x-1}$$


Solution

The function name is f , the independent variable is x and the function rule is $\frac{x}{x-1}$. So the function can be defined using one of the following methods.

- Method 1:
`f:=x->x/(x-1)`
- Method 2:
`f(x):=x/(x-1)`
- Method 3:
`f:=unapply(x/(x-1), x)`

Once the function is defined, you may find the function value using the following command.

`f(0.999)`

 **Exercise 1.3** Define the following function in Maple and find the value $f(2.0001)$.

$$g(x) = \frac{x^3}{(x-2)^2}$$

1.3.2 Initially known mathematical functions

Maple has many predefined functions which can be used to create new functions. To see all initially known mathematical functions in maple, you may use the help command `?functions` and click the hyperlinked “initial functions” in the description shown in the new window.

Functions

[Calling Sequence](#)
[Description](#)
[Examples](#)

Calling Sequence

`expr (expseq)`

Description

- This help page describes the Maple *function* expression. A “function” expression represents a function call, or application of a function or procedure to arguments. For a list of mathematical functions defined in Maple, see [initial functions](#). A mathematical function can be defined in Maple using a [functional operator](#). Use these to define a function of a single variable, a multivariate function, or a vector function.
- In Maple, a “function” expression represents a function call, or application of a function or procedure to arguments. Such an expression is said to be of type **function**. A “typical” example of an expression of type function is the expression $f(x)$, which represents the application of the expression f to the argument sequence x . Note that, in this expression, we refer to the entire expression $f(x)$ as being of type function (that is, a “function call” or “function application”), while the expression f is typically not itself of type function (but often is of type **procedure**).
- The operands of a function expression are the “arguments”. For example, the operands of the expression $f(a, b, c)$ are a , b and c . The expression f may be extracted from this expression as the zero-th operand. See the examples section below for more about this.

Maple function help page screenshot

Some frequently used functions are listed in tables below.

absolute value	square root	n-th root	natural exponential	logarithmic
<code>abs()</code>	<code>sqrt()</code>	<code>surd(,n)</code>	<code>exp()</code>	<code>log()</code> , <code>log[b]()</code> , <code>ln()</code>

sine	cosine	tangent	cotangent	secant	cosecant
<code>sin()</code>	<code>cos()</code>	<code>tan()</code>	<code>cot()</code>	<code>sec()</code>	<code>csc()</code>

inverse sine	inverse cosine	inverse tangent	inverse cotangent	inverse secant	inverse cosecant
<code>arcsin()</code>	<code>arccos()</code>	<code>arctan()</code>	<code>arccot()</code>	<code>arcsec()</code>	<code>arccsc()</code>

1.3.3 Evaluation and Substitution

To evaluate an expression with given values for the variables, there are multiple approaches.

- The `subs` command.
- Define a function using the expression and evaluate it using the function notation. For example, you may find the value of $e^{1.2}$ using the command `exp(1.2)`. Here `exp()` is the exponential function with the base e .

Example 1.3

The following codes show how to evaluate an expression using the `subs` command.

```
f:=a*x^2+b*x+c;
g:=subs({a=1, b=2, c=3}, f);
h:=subs(x=1, g);
```

Example 1.4


The following codes show how to evaluate an expression using the functional approach.

```
f := x -> x^2 + 2*x + 3;
f(1/2);
```

To evaluate expressions numerically, you need apply the command `evalf`. To keep only n digits in total, you may use `evalf[n]` or `evalf(, n)`.

Example 1.5 The command `evalf(sqrt(2))` will return the numerical value 1.414213562 of $\sqrt{2}$. The command `evalf[5](Pi)` will return the numerical value 3.1416 of π .

1. In Maple, the name `Pi` is for calculation. But the name `pi`, where `p` is in lower case, is for the mathematical constant π . You will see the difference when evaluating `sin(Pi)` and `sin(pi)`.
2. In Maple, to keep n decimal places of number, you may use `floor()+evalf[3](frac())`.

 **Exercise 1.4** Evaluate the expression $\sin(x) - 2x^2 - 1$ at $x = \pi$ and find the numerical value of the result.

1.4 Packages and Plotting

1.4.1 Frequently used packages and how to load them

A package is a collection of commands that extends the basic functionality of Maple and provided tools for solving problems of certain type or in certain field.

The package `Student` contains subpackages designed for learning of standard undergraduate mathematics courses, such as Calculus, Linear Algebra, Ordinary Differential Equations.

A few useful packages for differential equations are `Student[ODEs]`, `DETools` and `Plots`.

Packages can be loaded from the Menu bar or using the command `with()`. For example, the follows commands will load the above mentioned three packages without display the available commands of each package.

```
with(Student[ODEs]): with(DETools): with(plots):
```

Note that you may use the semicolon separator to see possible commands supported by a package.

1.4.2 Plot explicit functions

In Maple, you may plot a single variable function easily using the command

```
plot(expression, range, options)
```

or plot several single variable functions together using

```
plot([expression1, expression2], range, options)
```

In the command, options may be omitted, but the range must be given. A range of values for a variable in Maple has the form $x = a..b$, where a and b are the left and the right ends of the interval $[a, b]$ for x . For example, if the range $x = 0..5$ is given, the graph will be only plotted over the domain $[0, 5]$ for the function.


To see details about available options, you may run the command `?plot` in Maple.

Example 1.6 Plot the functions $f(x) = x^2$ in red and $l(x) = 2x + 1$ in blue over the domain $[-1, 2]$.

Solution Use the `plot` command to create graphs.

```
plot([x^2, 2*x+1], x=-1..2, color=[red, blue])
```


The output is shown below.



Plotting of two functions x^2 and $2x + 1$

The red curve is the graph of $y = x^2$. The blue curve is the graph of $y = 2x + 1$.

In this example, we used square brackets to order lists of functions and colors so that a function is paired with a color in the given order.

 **Exercise 1.5** Plot the functions $f(x) = \ln(x + 5)$ and $g(x) = 3\cos(2x + 1) + 4$ over the domain $[-\pi, \pi]$.

1.4.3 Plot implicit functions


To plot the curve of implicit functions defined by equations, we need the command `implicitplot` which is supported by the package `plots`. The usage of `implicitplot` is similar to the commands `plot`. Indeed, they share many common options. For details, you can run the command `?implicitplot`.

Example 1.7 Plot the curve defined by $x^2 + 4y^2 = 25$ in the region where $-5 \leq x \leq 5$ and $-3 \leq y \leq 3$.

Solution After loading the package, we may use this command to plot the graph.

```
with(plots):
implicitplot(x^2+4*y^2=25, x=-5..5, y=-3..3);
```

The output is the following graph.




An ellipse in squarely scaled coordinates

You will find the the graph looks a circle instead of a ellipse that the equation defined. This is because, the x -axis and y -axis are not in the same scaling. To have the same scaling, we need to add the option `scaling = constrained`.


```
implicitplot(x^2+4*y^2=25, x=-5..5, y=-3..3, scaling = constrained);
```

Here is the output:



An ellipse in equally scaled coordinates

Now the graph is in equally scaled coordinates.

 **Exercise 1.6** Plot the curve defined by $x^2 - 4y^2 = 9$ in the equally scaled region: $-5 \leq x \leq 5$ and $-3 \leq y \leq 3$.

1.5 Differentiation and Integration

- The `int(f(x), x)` calling sequence computes an indefinite integral $\int f(x)dx$. The `int(f(x), x=a..b)` calling sequence computes an indefinite integral $\int_a^b f(x)dx$.
- The `diff` command computes the partial derivative of the expression `a` with respect to x_1, x_2, \dots, x_n , respectively. The most frequent use is `diff(y(x), x)`, which computes the derivative $\frac{dy}{dx}$.
- The `implicitdiff(f(x,y)=g(x, y), y, x)` calling sequence computes the derivative $\frac{dy}{dx}$ using the implicit differentiation method.
- The differential operator `D` in maple applies to a function and returns a first order (partial) derivative as a function. For example, given a function $y = f(x, y)$, the command `D[1](f)` returns the partial derivative $\frac{\partial}{\partial x} f$.
- In Maple (version ≥ 10), the prime derivative notation also works. For example, given a function `f(x) := sin(x^2)`, the second derivative of f can be calculated using the command `f''(x)`. When applying the prime derivative notation to a function or an expression, the result is in the same type. The default differentiation variable for the prime notation is x . To change it to another variable, say t , you may use the following command

```
Typesetting:-Settings(prime = t)
```

```
# Change the default differentiation variable to t.
```

Here, the symbol `#` is the comment symbol in Maple.

When using `diff` or the prime derivative notation to an expression, the result is an expression. To evaluate it, you may use the `subs` command. Another way is to apply `unapply` to the result to convert it into a function (operator).

Example 1.8 Find $f''(2)$ for the function $f(x) = \sin(x - 1)$.

Solution

We first define a function f .

```
f(x):=sin(x-1)
```

To calculate the value of the second derivative $f''(2)$, the easiest way probably is to use the prime notation.

```
f''(2)
```

If you used *diff* to find the derivative, you may run the following command to get the value $f''(2)$

```
unapply(diff(f(x), x$2), x)(2)
```

Using the command D , you may use the following command

```
D(D(f))(2) # D^((2))(f)(2)
```

In this example, the sequence operator $\$$ is used. The command $x\$2$ is equivalent to x, x .

Exercise 1.7

Use above commands to solve the following problems.

1. Find the indefinite integral of the function $\frac{1}{x}$.
2. Find the first derivative of the function $\arctan(x)$.
3. Find $\frac{dy}{dx}$, where x and y satisfy the equation $x^2 - 4y^2 = 9$.

Chapter 2 Introduction to Differential Equations

2.1 Basic concepts

2.1.1 How to define a differential equation in Maple

Depending on the differentiation command, there are three ways to define a differential equation in Maple.

Example 2.1 Assign the differential equation $y' = 2y$ to a variable in Maple.

Solution

- Method 1: Using the *diff* command
`ode111:=diff(y(x), x)=2*y(x)`
- Method 2: Using the the prime derivative notation
`ode112:=y'=2y`
- Method 3: Using the command *D*
`ode113:=D(y)(x)=2*y(x)`

Among the three methods, the one using *diff* is the standard choice.

 **Exercise 2.1** Assign the differential equation $y' = y(1 - 0.001y)$ to a variable in Maple.

2.1.2 How to check solutions

To check if a function (explicitly or implicitly defined) is a solution of a given differential equation, you may use the `odetest(function, ODE, y(x))` command. If the output is 0, then the function is a solution.

Example 2.2 Verify that $y = ce^{2x}$ is a solution to the differential equation $y' = 2y$.

Solution

Run the following command, you will see that output is 0. So the function is a solution.

```
odetest(y(x)=c*exp(2*x), diff(y(x), x)=2*y(x), y(x))
```


Example 2.3 Verify that $x^2 + y^2 = 1$ is a solution of the equation $y' = -\frac{x}{y}$.


Solution

Again, running the following command returns the number 0. So the implicit function is a solution.

```
odetest(x^2+y(x)^2=1, diff(y(x),x)=-x/(y(x)), y(x));
```

When working with differential equations, we should always use $y(x)$ instead of y to indicate that y is a function of x .

 **Exercise 2.2** Verify that the function $y = c_1 e^x + c_2 e^{-x}$ is a solution of the equation $y'' = y$.

 **Exercise 2.3** Verify that $x^2 + 4y^2 = c$ is an implicit solution of the equation $4yy' = x$.

2.2 Solution curves vs Integral curves

A solution curve is the graph of a function $y = f(x)$ that satisfies the given differential equation. An integral curves is a union of solution curves.

Example 2.4 Consider the differential equation $yy' = 4x$. Verify that the graphs of the functions $y = \pm\sqrt{4x^2 - 1}$ are solution curves of the equation, while the hyperbola $4x^2 - y^2 = 1$ defines an integral curve of the equation.

Solution

Let's rename the function as $y_1(x) = \sqrt{4x^2 - 1}$ and $y_2(x) = -\sqrt{4x^2 - 1}$. In Maple, it means we $y[1](x)$ and $y[2](x)$. We can check that they are solutions using the `seq` loop.

```
ode121:=y(x)*diff(y(x), x)=4*x:
y[1](x):=sqrt(4*x^2-1);
y[2](x):=-sqrt(4*x^2-1);
seq(odetest(y(x)=y[i](x), ode121, y(x)), i = 1 .. 2);
```

The outputs show that y_1 and y_2 are solutions.

To see that hyperbola $4x^2 - y^2 = 1$ defines an integral curve, we solve for y .

```
soly:=solve(4*x^2-y^2=1, y);
```


You will see that the solutions are exactly the functions y_1 and y_2 . So as an union of solution curves the hyperbola is an integral curve.

Plotting those curves will help use understand better.

```
solutioncurves := plot([y[1](x), y[2](x)],
  x = -5 .. 5, y = -5 .. 5, color = [green, red]);
with(plots):
integralcurve := implicitplot(4*x^2 - y^2 = 1,
  x = -5 .. 5, y = -5 .. 5, color = blue,
  linestyle = dot);
```


To check the integral curve is the union of the two solution curve, we can use the *display* command.

```
display(solutioncurves,integralcurve)
```

 **Exercise 2.4** Consider the differential equation $yy' = 4x$. Verify that the graphs of the functions $y = \pm\sqrt{4x^2 + 1}$ are solution curves of the equation, while the hyperbola $y^2 - 4x^2 = 1$ defines an integral curve of the equation.

2.3 Direction fields

In Maple, the commands `DEplot`, `dfieldplot`, and `phaseportrait` supported by the package `DETools` can be used to plot the direction field and solution curves. The basic usage is as follows

```
DEplot(differential equation, function, ranges, options)
```

Again, using the command `?DEplot`, we can find details and examples on the command. In the following, I will use `DEplot` as an example to show how they work.

Example 2.5 Plot the direction field for the differential equation $y' = -\frac{x}{y}$. Can you guess what is a solution to this equation?

Solution Let's first define the differential equations and assign it to a variable.

```
ode131:=diff(y(x), x)=-x/(y(x));
```

Now load the package *DETools* using the command *with()*.

```
with(DETools):
```

With the package loaded, we can use *DEplot* to plot the direction field for *ode131*, say in the region $-5 \leq x \leq 5$ and $-5 \leq y \leq 5$.

```
DEplot(ode131, y(x), x = -5..5, y = -5..5,
      title = " Direction Field for y'=-x/y ")
```

Note that one may change the outlook by add options, such as color and arrows. Running the following command, you will see the difference.

```
DEplot(diff(y(x), x) = -(x - 1)/(y(x) + 1), y(x),
      x = -5 .. 5, y = -5 .. 5,
      title = "Direction Field for y'=2 y",
      color = -(x - 1)/(y(x) + 1), arrows = line)
```

The direction field suggests that solutions are circles.

To display $y(x)$ as y in the output, you may run the following commands first.

```
with(PDEtools, declare):
```

```
declare(y(x), prime=x); # Turn ON the enhanced DEdisplay feature
```

You will see the following output

derivatives with respect to x of functions of one variable will now be displayed with '

🔗 **Exercise 2.5** Plot the direction field for the differential equation $y' = -\frac{x-1}{y+1}$. Can you guess what is a solution to this equation?

🔗 **Exercise 2.6** Plot the direction field for the differential equation $y' = -\frac{x}{2y}$. Can you guess what is a solution to this equation?

Chapter 3 First Order Differential Equations

3.1 Classification

Knowing the type of a differential equation will be very helpful for solving it. In Maple, we can use the command `odeadvisor(ode, y(x))` supported by the package `DETools` to learn what type of equation is it.

Example 3.1

Determine the primary type of each of the following differential equations.

1. $y' = xy$
2. $y' = xy + 1$
3. $y' = xy + y^2$
4. $y' = \frac{xy}{x^2 + y^2}$

Solution

We first load the package *DETools* anonymously.

```
with(DETools):
```

Now we can classify those equations using *odeadvisor*.

```
odeadvisor(diff(y(x),x)=x*y(x));  
odeadvisor(diff(y(x),x)=x*y(x)+1);  
odeadvisor(diff(y(x),x)=x*y(x)+y(x)^2);  
odeadvisor(diff(y(x), x) = (x*y(x))/(x^2 + y(x)^2));
```

Exercise 3.1

Determine the primary type of each of the following differential equations.

1. $y' = x^2y^2 + x^2$
2. $y' = x^2y + x$
3. $y' = xy + y^3$
4. $y' = \frac{x-y}{x+y}$

3.2 Solving differential equations

To solve an ordinary differential equation, or a system of them, or initial value problems, you may use the command `dsolve({ODE, InitialConditions}, y(x), options)`, where the

initial conditions may be omitted to get a general solution. Note that `dsolve` returns an equation with $y(x)$ on the left.

Among options, you may choose to use different method to solve the equations, for example, `numeric`, `series` or `method=laplace` are options that can be imposed and will be used later.

Example 3.2

Consider the differential equation $y' = 2y + x$.

1. Find the general solution.
2. Find the solution that satisfies the initial condition $y(0) = 1$.
3. Plot the solution curve of the initial value problem.

Solution

It will be convenient to define the differential equation first.

```
ode221:=diff(y(x), x)=2*y(x)+x:
```

The general solution can be obtained by the following command.

```
dsolve(ode221, y(x));
```

The solution of the initial value problem can be obtained by

```
sol221:=dsolve({ode221, y(0)=1}, y(x));
```

To plot the solution curve, we need to get the function expression instead of the equation. This can be done using the command `rhs`.

```
plot(rhs(sol221), x=-5..5)
```

By default, the `dsolve` command only returns a function. To show more detailed information about the computation, you may assign to `infolevel[dsolve]` an integer 1 through 5 before using the `dsolve` command. For example, after running the following command `infolevel[dsolve]:=3`, the output from `dsolve` will show three pieces of information: methods tried, successful or not, the solution.

Exercise 3.2

Consider the differential equation $y' = x + y$.

1. Find the general solution.
2. Find the solution that satisfies the initial condition $y(0) = 1$.
3. Plot the solution curve of the initial value problem.

3.3 Integrating factors

In Maple, you may use `intfactor` that is supported by the package `DETools` to find an integrating factor for a given ODE. To test an integrating factor, you may use `mutest` which is again supported by the package `DETools`. If the command returns a 0, then the expression being tested is an integrating factor.

Example 3.3 Find an integrating factor for the equations $y' = x^2y - x$ and test it using `mutest`.

Solution

Let's first define the differential equation.


```
ode231:=diff(y(x), x)=x^2*y(x)-x:
```

Now loading the package and find an integrating factor μ .

```
with(DETools):
mu:=intfactor(ode231);
```

To test it, you may run the following command.

```
mutest(mu, ode231, y(x));
```

 **Exercise 3.3** Find an integrating factor for the differential equation. $xy' = y - x$.

3.4 Commands for specific type of equations

The `DETools` package also provides some commands for solving specific type of equations. The following is an incomplete list.

1. The command `separablesol(ode, dependent variable)` determines whether the ode is a separable first order ODE and, if so, returns a solution to the equation.

Example 3.4 Consider the first order differential equation

$$y' = \frac{x}{x^2y + y}.$$

Executing The following codes

```
with(DEtools):
sode := diff(y(x), x) = x/(x^2*y(x)+y(x));
exactsol(sode, y(x));
```

will produce a set of solutions

$$\left\{ y(x) = \sqrt{\ln(x^2 + 1) - 2_C1}, y(x) = -\sqrt{\ln(x^2 + 1) - 2_C1} \right\}.$$

2. The command `linearsol(ode, dependent variable)` determines whether the ode is a first order linear ODE and, if so, returns a solution to the equation.

Example 3.5 Consider the linear first order differential equation

$$y' + p(x)y = q(x).$$

Executing The following codes

```
with(DEtools):
lode := diff(y(x), x) + p(x)*y(x) = q(x);
linearsol(lode, y(x));
```

will produce a set of solutions

$$\{y(x) = \int (-p(x)z(x) + q(x))dx + _C1\}$$

3. The command `exactsol(ode, dependent variable)` determines whether the ode is an exact first order ODE and, if so, returns a solution to the equation. When the equation is not exact, it tries to find an integrating factor that converts the equation into an equivalent exact equation.

Example 3.6 Consider the first order differential equation

$$(x^2 - y^2)dx + 2xydy.$$

Executing The following codes

```
with(DEtools):
eode := x^2 - y(x)^2 + 2*x*y(x)*diff(y(x), x) = 0;
exactsol(eode, y(x));
```

will produce a set of solutions

$$\{y(x) = \sqrt{-_C1x - x^2}, y(x) = -\sqrt{-_C1x - x^2}\}$$

One can also specific the method (type of the equation) in the `dsolve` command so that the equation will be solve by using the given method.

Chapter 4 Linear Second Order Differential Equations

4.1 The general solution of a linear equation

Consider the linear second order differential equation

$$y'' + p(x)y' + q(x)y = f(x).$$

Let y_p be a particular solution, and y_h the general solution of the associated homogeneous equation

$$y'' + p(x)y' + q(x)y = 0.$$

Then the function $y_h + y_p$ is the general solution of the non-homogeneous equation $y'' + p(x)y' + q(x)y = f(x)$.

The function y_h can be written as $c_1y_1 + c_2y_2$, where y_1 and y_2 are linearly independent solutions of the associated homogeneous equation. To find y_1 and y_2 , we can use `dsolve` with the option `output=basis`.

Example 4.1 Find two linearly independent solutions of the equation

$$y'' + y = 0.$$

Solution We first run commands which provide better display.

```
PDETools[declare](y(x), prime=x): # Turn ON the enhanced DEDisplay feature
infolevel[dsolve] := 3: # Show more detailed information about computation
```

Now define the equation

```
ode31:=diff(y(x), x$2)+y(x)=0 # here x$n = x, x, ..., x, n copies of x
```

To find the independent solutions, run the following command

```
dsolve(ode31, y(x), output=basis)
```

The output is

$$[\sin(x), \cos(x)]$$

which means that $\sin x$ and $\cos x$ are linearly independent solutions.

If the homogeneous linear equation has constant coefficients, then one can also use `DETools[constcoeffsols]` to find a list of independent solutions.

 **Exercise 4.1** Find two linearly independent solutions of the equation

$$y'' - y' - 2y = 0.$$

To find a particular solution of a nonlinear ODE, or a linear non-homogeneous ODE, one can use the command `particularsol(ODE, dependent variable)`.

Example 4.2 Find a particular solution to the equation

$$y'' - 3y' + 2y = 2.$$

Solution

First define the equation


```
ode32:=diff(y(x), x, x)-3*diff(y(x), x)+2*y(x)=2
```

To find a particular solution, we can use

```
with(DETools):
particularsol(ode32, y(x));
```

or

```
DETools[particularsol](ode32, y(x))
```

 **Exercise 4.2** Find a particular solution to the equation

$$y'' - 3y' + 2xy = 2x^2 - 3.$$

4.2 Wronskian of Solutions

The Wronskian of two solutions y_1 and y_2 of a linear homogeneous second order differential equation $y'' + p(x)y' + q(x)y = 0$ is defined by

$$W(y_1, y_2) = y_1 y_2' - y_1' y_2$$

. In Maple, one can use the command `Wronskian([y_1, y_2], independent variable, determinant=true)` or `LinearAlgebra[Determinant](Wronskian([y_1, y_2], independent variable))` to find the Wronskian.

Example 4.3 The equation $x^2 y'' + 3xy' - 3y = 0$ has two solutions $y_1 = x$ and $y_2 = x^{-3}$. Find the Wronskian and determine if the two solutions are linearly independent.

Solution We first load the package *VectorCalculus*.

```
with(VectorCalculus):
```


Now we calculate the Wronskian.

```
LinearAlgebra[Determinant](Wronskian([x, x^(-3)], x))
```

The output is

$$-\frac{4}{x^3}.$$

Because the Wronskian is not identically zero, the solutions are linearly independent.

 **Exercise 4.3** The equation $x^2y'' + 4xy' - 4y = 0$ has two solutions $y_1 = x$ and $y_2 = x^{-4}$. Find the Wronskian and determine if the two solutions are linearly independent.

4.3 Linear Equations with Constant Coefficients

When solving linear equations with constant coefficients, we need to solve equations and system of equations. That can be done by the Maple command `solve({equations}, {unknowns})`. Moreover, one can add constraints by using `assuming` together with `solve`.

Example 4.4 Solve the initial value problem

$$y'' - 2y' - 3y = 0, \quad y(0) = 1, y'(0) = -1.$$

Solution

Since this equation is a linear second order equation with constant coefficient, the solution is determined by the roots of the characteristic equation $r^2 - 2r - 3 = 0$. Running the following Maple command will give us the roots.

```
rts:=solve(r^2-2*r-3=0, r);
```

The roots are

$$3, -1.$$

Since the roots are real numbers, the general solution of the differential equation is

```
yh(x):=c1*exp(rts[1]*x)+c2*exp(rts[2]*x);
```

The initial conditions impose two equations for the constants `c1` and `c2`. We can calculate y' using $D(y)$ and solve the constants by the following command

```
consts:=solve({yh(0)=1, D(yh)(0)=-1}, {c1,c2});
```

The output is

$$\{c1 = 0, c2 = 1\}.$$


So the solution to the initial value problem is

$$y(x) = e^{-x}$$

which can be seen by the Maple commands

```
y(x):=subs(consts, yh(x));
y(x);
```

Here, we take the approach of solving the problem manually. In Maple, we can use `dsolve` or `constcoeffsols` to solve differential equations with constant coefficients.

 **Exercise 4.4** Solve the initial value problem

$$y'' + 3y' + 2y = 0, \quad y(0) = 3, y'(0) = 1.$$

4.4 Undetermined Coefficients

For some equations that can be written into the form

$$ay'' + by' + cy = f(x)$$

, a particular solution can be found using the method of undetermined coefficients. Calculations can be made easy using Maple.

Example 4.5 Find a particular solution of

$$y'' - 5y' + 6y = 3e^{2x}.$$

Solution Since the right hand side is an exponential function, and 2 is a root of the characteristic polynomial of the complementary equation, we expect a specific solution $y_p = Axe^{2x}$. We then plug it into the equation to solve for A .

Define the differential equation and the solution function.


```
ode34:=diff(y(x), x, x)-5*diff(y(x), x)+6*y(x)=3*exp(2*x);
yp(x):=A*x*exp(2*x);
```

Plugging y_p into the differential equation and solve for A .

```
eqnA:=subs[eval](y(x)=yp(x), ode34);
A=solve(eqnA, A);
```

The output is

$$A = -3.$$

 **Exercise 4.5** Find a particular solution of

$$y'' - 5y' - 6y = 2e^{3x}.$$

4.5 Reduction of Order

Given a solution y_1 of a linear second order differential equation $y'' + p(x)y' + q(x)y = f(x)$, one can use the trick of variation of parameter to find another solution in the form $y_2 = y_1 v$, here v satisfies a second order equation in the form $y_1 v'' + v'(p(x)y_1 + 2y_1) = 0$. This equation can be solve by reduction of order using the substitution $u = v'$. Indeed, we can expect $y_2 = y_1 \int u(x)dx$.

In Maple, one can use the command `reduce_order(ode, y(x)=y_1(x), u(t))` to obtain a lower order equation for u . This command again belongs to the package `DETools`.

Example 4.6 The equation $x^2 y'' + 4xy' - 4y = 0$ has a solution $y_1 = x$. Find another linearly independent solution y_2 .

Solution

Let's load packages first. Since *with* can only load one package each time, let's use a *for* loop to load packages. To make the display looks better, let's also run the *declare* command.

```
for i in [DETools, LinearAlgebra, PDETools] do with(i) end do;
declare(y(x), prime=x):
```

Define the equation

```
ode35:=x^2*diff(y(x), x, x)+4*x*diff(y(x),x)-4*y=0;
```

Find the equation that u satisfies.

```
RdOrder:=reduce_order(ode35, y(x) = x, u(t));
```

The output is

$$y = \left(t \left(\int u(t) dt + _C1 \right) \right) \text{ where } \left[\{t(u_t) + 6u(t) = 0\}, \left\{ t = x, u(t) = \frac{x(y') - y}{x^2} \right\}, \left\{ x = t, y = t \left(\int u(t) dt + _C1 \right) \right\} \right]$$

You can see that the equation for u is $t(u_t) + 6u(t) = 0$, where u_t is $v'(t)$.

We can solve for u and then find v . Here, we use the operand extract command *op* to get the differential equation for u . The equation is the first element in the first list of the second vector of the right hand side of the output. We now get the equation for $u(x)$.

```
odeu:=subs(t=x, op([2, 1, 1], rhs(RdOrder)));
```

Now we substitute t by x and solve for u and v .

```
ux:=dsolve(odeu, u(x));
```

```
v:=int(ux, x);
```


Therefore, by the following command, we know that $y_2(x) = \frac{c}{x^4}$.

```
y[2](x):=x*v(x);
```

Testing y_2 using the following command shows that it is a solution.

```
odetest(y(x)=y[2](x), ode35);
```

- To get the second solution y_2 after obtained the differential equation for u , one can also use the command
`buildsol(Reduced_Order_Equation, Particular_Solution_of_u)`
- One can also find y_2 without using `DETools` (see for example the Maple document on Linear Equations created by Prof. Douglas B. Meade.)

 **Exercise 4.6** The equation $x^2y'' + 3xy' - 3y = 0$ has a solution $y_1 = x$. Find another linearly independent solution y_2 .

4.6 Variation of Parameters

Maple has the command `varparam(Solutions, f(x), x)` to find the general solution of a linear ODE by the method of variation of parameters, where $f(x)$ is the right-hand side function. Again this command belongs to `DETools`.

Example 4.7 Find the general solution for the equation

$$x^2y'' + 4xy' - 4y = x$$

using the solutions $y_1 = x$ and $y_2 = \frac{1}{x^4}$ of the complementary equation

$$x^2y'' + 4xy' - 4y = 0.$$

Solution We can find the general solution using one line of command.

```
DETools[varparam]([x, 1/x^4], x, x)
```

The output is

$$-C_1x + \frac{C_2}{x^3} + \frac{x^3}{14}$$

 **Exercise 4.7** Find the general solution for the equation

$$x^2y'' + 3xy' - 3y = x$$

using the solutions $y_1 = x$ and $y_2 = \frac{1}{x^3}$ of the complementary equation

$$x^2y'' + 3xy' - 3y = 0.$$